Máster Universitario en Ingeniería de Caminos, Canales y Puertos
Departamento de Matemática Aplicada
Análisis Numérico

Práctica 4
Resolución de sistemas de ecuaciones no lineales.

Introducción

En esta parte de la práctica, veremos el método de Newton-Raphson y de iteración funcional para varias variables.

El método de Newton-Raphson multivariable

```
Clear["Global`*"]
```

Definimos una manera de hallar la norma de un vector que usaremos después en el criterio de parada.

```
Norma[a_, b_] := Sqrt[a^2 + b^2]
```

Definimos la función de dos variables.

```
F[{x_, y_}] := { x^2 + y^2 - 4, x * y - 1}
```

Comprobamos la definición.

```
F[{1, 2}]
```

Para hallar la matriz jacobiana necesitaremos las derivadas de las componentes de F. Vemos antes una manera de obtener las componentes de F.

```
F[{x, y}][[1]]
```

Por simplificar, definiremos aparte las dos componentes.

```
f1[x_, y_] := x^2 + y^2 - 4
```

Usamos el comando D para derivar. Vemos como se usaría para hallar la derivada con respecto a x de la primera componente de F.

```
D[f1[x, y], x]
```

```
D[F[{x, y}][[1]], x]
```

Definimos la segunda componente de F.

```
f2[x_, y_] := x * y - 1
```

Definimos la matriz jacobiana que contiene las derivadas de las componentes de F.

```
J[x_, y_] := {{ D[f1[xx, yy], xx], D[f1[xx, yy], yy]},
   {D[f2[xx, yy], xx], D[f2[xx, yy], yy]}} /. {xx -> x, yy -> y}
```

En esta definición de J, los valores de x e y son números en los que se evalúa la matriz, por tanto no podemos derivar f1[x,y] respecto de x. Usaremos las variables mudas xx e yy, para después sustituir xx por el valor concreto de x, y sustituir yy por el valor de y. La sustitución se hace con /.{xx->x,yy->y}

Comprobamos la definición de J.

```
J[x, y]
```

Definimos la matriz inversa de la matriz jacobiana. En este caso sencillo es posible hacerlo de forma simbólica antes del bucle, en otros casos habría que resolver un sistema lineal en cada iteración.

```
Inv[x_, y_] := Inverse[J[x, y] ]
```

```
Inv[x, y]
```

Inicializamos el algoritmo.

```
x0 = 0.0;
y0 = 1.0;
tol = 0.0001;
seguir = True;
```

A continuación el bucle del método.

```
While[seguir == True,
    {x1, y1} = {x0, y0} - Inv[x0, y0].F[{x0, y0}];
    Print["x1=", x1, "    y1=", y1];
    If[ Norma[x0 - x1, y0 - y1] < tol, seguir = False];
    x0 = x1;
    y0 = y1
    ]
```

Comprobamos la solución obtenida.

```
F[{x0, y0}]
```

El método de Punto Fijo multivariable

```
Clear["Global`*"]
```

Definimos una norma que se aplique a vectores.

```
Norma[{a_, b_}] := Sqrt[a^2 + b^2]
```

Definimos la función F, que usaremos para comprobar el método, pero que no se usa en la implementación.

```
F[{x_, y_}] := { x^2 + y^2 - 4, x * y - 1}
```

Definimos la función G, que se ha obtenido despejando x de la primera ecuación e y de la segunda ecuación. Nótese que el argumento de G es un vector con dos componentes, y la imagen de G es un vector de dos componentes. Esto facilitará la iteración de G.

```
G[{x_, y_}] := {Sqrt[4 - y^2], 1 / x}
```

Definimos un vector v0, e inicializamos las variables seguir y tol.

```
v0 = {1.0, 1.0};
seguir = True;
tol = 0.000001;

While[seguir == True,
    v1 = G[v0];
    If[Norma[v1 - v0] < tol, seguir = False];
    v0 = v1
]

F[v1]
```

COMANDOS

D[función, variable]

Efectúa la derivada de la función respecto de la variable. Puede usarse para derivadas sucesivas.

```
D[Sin[7 * x],
 x]
```

7 Cos[7 x]

```
D[x^5, {x,
  2}]
```

$20\,x^3$

EJERCICIOS

1.- Aplique el método de Newton-Raphson para encontrar dos soluciones del sistema

x^3+y^2=2,

x^2+2y+2=0.

(Indicación: comience por {1, -1}, y por {-3, -6})

(Sol. {0.732051, -1.26795}, {-2.73205, -4.73205})

2.- Modifique el método de Newton-Raphson para que no sea necesario definir la inversa de matriz jacobiana antes de comenzar el bucle, sino que en cada iteración se resuelva el sistema lineal pertinente, usando la orden LinearSolve.

3.- Para la función del ejemplo de punto fijo. ¿Puede usarse como valor inicial {0, 1.0}? ¿y {0.1, 1.0}?

4.- Encuentre una solución del sistema

x^2 + 2 y^2 + 6 x - 3 = 0,

x^2 + y^2 + 5 y - 3 == 0

partiendo de la solución inicial {-1.5, -1.5}

(Indicación: despeje la x de 6x en la primera ecuación y la y de 5y en la segunda).

(Sol. {0.386119,0.516758})